
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:
Shaun Harry

shaun.harry@keysight.com
2 Corrections required
2.1 Change 1

	Function name
	fl_TC_8_1_4_2_2_1_NR5GC_EUTRA_TestBody

	Reason for change
	According to 33.501 s8.4:
……..

2.The source MME selects the target AMF and sends a Forward Relocation Request to the selected target AMF. The source MME includes UE's EPS security context including KASME, eKSI, UE EPS security capabilities, selected EPS NAS algorithm identifiers, uplink and downlink EPS NAS COUNTs, {NH, NCC} pair, in this message. If the source MME has the UE NR security capabilities stored, then it will forward the UE NR security capabilities as well to the target AMF.

3.
The target AMF shall construct a mapped 5G security context from the EPS security context received from the source MME. The target AMF shall derive a mapped KAMF key from the received KASME and the NH value in the EPS security context received from the source MME as described in clause 8.6.2.

…….

So, for EUTRA->NR handover, EUTRA PTC should send {NH, NCC} to NR PTC for creating mapped 5G  security context
According to TS 33.401 7.2.8.1.1, NH associated to NCC=1 is not used to derive keys, so {NH, NCC=2} is used


	Summary of change
	Derive NH corresponding to NCC=2, and add {NH, NCC=2} to EUTRA_NR_Coordination_MSG

	TTCN module
	NR5GC_IRAT_CommonFunctions

	MCC160 Comment
	Accepted, but NCC value should be incremented by 1 instead of hardcoded to 2, according to 33.401 clause 7.2.8.4.3


Before Change

	  function fl_TC_8_1_4_2_2_1_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var octetstring v_NR_HOCmd;

    //@siclog "Step 1" siclog@

    // Send latest parameters to NR

    f_EUTRA38_Send_AuthParametersToNR();

    f_EUTRA38_SendPduSessionInfoListToNR();

    // Wait for NR to send back encoded RRC Reconfiguration message

    v_NR_HOCmd := f_EUTRA_NR_WaitForCoOrd_Data(NR, "HO Cmd");

    SRB.send( cas_SRB1_RrcPdu_REQ( eutra_Cell1, cs_TimingInfo_Now, cs_MobilityFromEutraCommand_Handover( tsc_RRC_TI_Def,

                                                                                                         false,

                                                                                                         nr,

                                                                                                         v_NR_HOCmd,

                                                                                                         omit)));

    // UE now moved to NR

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

  }


After Change

	function fl_TC_8_1_4_2_2_1_NR5GC_EUTRA_TestBody() runs on EUTRA_5GS_PTC

  {

    var octetstring v_NR_HOCmd;


  var EUTRA_SecurityParams_Type v_Auth_Params;

    v_Auth_Params := f_EUTRA_Security_Get();

    v_Auth_Params.NH := f_EUTRA_NB_Authentication_S12(v_Auth_Params.KDF, v_Auth_Params.KASME, v_Auth_Params.NH);  


  v_Auth_Params.NCC := 2;


  f_EUTRA_Security_Set(v_Auth_Params);
    //@siclog "Step 1" siclog@

    // Send latest parameters to NR

    f_EUTRA38_Send_AuthParametersToNR();

    f_EUTRA38_SendPduSessionInfoListToNR();

    // Wait for NR to send back encoded RRC Reconfiguration message

    v_NR_HOCmd := f_EUTRA_NR_WaitForCoOrd_Data(NR, "HO Cmd");

    SRB.send( cas_SRB1_RrcPdu_REQ( eutra_Cell1, cs_TimingInfo_Now, cs_MobilityFromEutraCommand_Handover( tsc_RRC_TI_Def,

                                                                                                         false,

                                                                                                         nr,

                                                                                                         v_NR_HOCmd,

                                                                                                         omit)));

    // UE now moved to NR

    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Postamble");

  }


2.2 Change 2
	Function name
	Typedef EUTRA_NR_CoOrd_Security_Type

	Reason for change
	See change 1

	Summary of change
	Extend record type to add {NH, NCC}

	TTCN module
	EUTRA_NR_CoordMsg

	MCC160 Comment
	Accepted


Before Change

	  type record EUTRA_NR_CoOrd_Security_Type {  /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT, POS) */

    // All fields are optional as some are only required for different options

    integer             SK_Counter  optional,

    CoOrd_KeyToUse_Type KeyToUse    optional,

    B256_Type           Key         optional,

    B3_Type             KSI         optional,

    NasCount_Type       NASCountUL  optional,   // @sic R5w190307 sic@

    NasCount_Type       NASCountDL  optional,   // @sic R5w190307 sic@

    CoOrd_E_NEACap_Type E_NEACap    optional,

    CoOrd_E_NIACap_Type E_NIACap    optional

  };


After Change

	  type record EUTRA_NR_CoOrd_Security_Type {  /* @status    APPROVED (ENDC, NR5GC, NR5GC_IRAT, POS) */

    // All fields are optional as some are only required for different options

    integer             SK_Counter  optional,

    CoOrd_KeyToUse_Type KeyToUse    optional,

    B256_Type           Key         optional,

    B3_Type             KSI         optional,

    NasCount_Type       NASCountUL  optional,   // @sic R5w190307 sic@

    NasCount_Type       NASCountDL  optional,   // @sic R5w190307 sic@

    CoOrd_E_NEACap_Type E_NEACap    optional,

    CoOrd_E_NIACap_Type E_NIACap    optional,


  B256_Type           NH          optional,


  integer             NCC         optional
  };


2.3 Change 3
	Function name
	Template cms_EUTRA_NR_Security

	Reason for change
	See change 1

	Summary of change
	Extend template to send {NH, NCC}

	TTCN module
	EUTRA_NR_CoordMsg

	MCC160 Comment
	Accepted


Before Change

	  template (value) EUTRA_NR_Coordination_MSG cms_EUTRA_NR_Security(template (omit) integer             p_SK_Counter,

                                                                   template (omit) CoOrd_KeyToUse_Type p_KeyToUse,

                                                                   template (omit) B256_Type           p_Key,

                                                                   template (omit) B3_Type             p_KSI,

                                                                   template (omit) NasCount_Type       p_NASCountUL,

                                                                   template (omit) NasCount_Type       p_NASCountDL,

                                                                   template (omit) CoOrd_E_NEACap_Type p_E_NEACap,

                                                                   template (omit) CoOrd_E_NIACap_Type p_E_NIACap) :=

  { /* @status    APPROVED (ENDC, NR5GC_IRAT) */

    // Added fields for IRAT @sic R5-194814 sic@

    Security := {

      SK_Counter := p_SK_Counter,

      KeyToUse   := p_KeyToUse,

      Key        := p_Key,

      KSI        := p_KSI,

      NASCountUL := p_NASCountUL,

      NASCountDL := p_NASCountDL,

      E_NEACap   := p_E_NEACap,

      E_NIACap   := p_E_NIACap

    }

  };


After Change

	template (value) EUTRA_NR_Coordination_MSG cms_EUTRA_NR_Security(template (omit) integer             p_SK_Counter,

                                                                   template (omit) CoOrd_KeyToUse_Type p_KeyToUse,

                                                                   template (omit) B256_Type           p_Key,

                                                                   template (omit) B3_Type             p_KSI,

                                                                   template (omit) NasCount_Type       p_NASCountUL,

                                                                   template (omit) NasCount_Type       p_NASCountDL,

                                                                   template (omit) CoOrd_E_NEACap_Type p_E_NEACap,

                                                                   template (omit) CoOrd_E_NIACap_Type p_E_NIACap,

                                                         
    template (omit) B256_Type           p_NH := omit,












    template (omit) integer             p_NCC := omit) :=

  { /* @status    APPROVED (ENDC, NR5GC_IRAT) */

    // Added fields for IRAT @sic R5-194814 sic@

    Security := {

      SK_Counter := p_SK_Counter,

      KeyToUse   := p_KeyToUse,

      Key        := p_Key,

      KSI        := p_KSI,

      NASCountUL := p_NASCountUL,

      NASCountDL := p_NASCountDL,

      E_NEACap   := p_E_NEACap,

      E_NIACap   := p_E_NIACap,


    NH         := p_NH,


    NCC        := p_NCC
    }

  };


2.4 Change 4
	Function name
	f_EUTRA38_Send_AuthParametersToNR

	Reason for change
	See change 1

	Summary of change
	Send {NH, NCC} to NR

	TTCN module
	EUTRA38_NR5GCProcedures

	MCC160 Comment
	Accepted


Before Change

	  function f_EUTRA38_Send_AuthParametersToNR () runs on EUTRA_5GS_PTC

  {

    var EUTRA_SecurityParams_Type v_NAS_SecurityParams := f_EUTRA_Security_Get();

    var NasCountInfo_Type v_NASCountInfo := f_EUTRA_NB_SS_NAS_CountGet(NASCTRL);

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Security(omit, omit,

                                                    v_NAS_SecurityParams.KASME,

                                                    v_NAS_SecurityParams.KSIasme,

                                                    v_NASCountInfo.UL,

                                                    v_NASCountInfo.DL, // @sic R5w190307 sic@

                                                    omit, omit));

  }


After Change

	  function f_EUTRA38_Send_AuthParametersToNR () runs on EUTRA_5GS_PTC

  {

    var EUTRA_SecurityParams_Type v_NAS_SecurityParams := f_EUTRA_Security_Get();

    var NasCountInfo_Type v_NASCountInfo := f_EUTRA_NB_SS_NAS_CountGet(NASCTRL);

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Security(omit, omit,

                                                    v_NAS_SecurityParams.KASME,

                                                    v_NAS_SecurityParams.KSIasme,

                                                    v_NASCountInfo.UL,

                                                    v_NASCountInfo.DL, // @sic R5w190307 sic@

                                                    omit, omit,

                                                    v_NAS_SecurityParams.NH,










 v_NAS_SecurityParams.NCC));

  }


2.5 Change 5
	Function name
	f_NR5GC_508RRC_InterRAT_HO_ToNR

	Reason for change
	According to 33.501 s8.4.2

The target AMF shall create a NAS Container to signal the necessary security parameters to the UE. The NAS Container shall include a NAS MAC, the selected 5G NAS security algorithms, the ngKSI associated with the derived KAMF' and the NCC value associated with the NH parameter used in the derivation of the KAMF'.  The target AMF shall calculate the NAS MAC as described in clause 6.9.2.3.3. with the COUNT parameter set to the maximal value of 232-1.

	Summary of change
	Use the ngKSI, NCC recieved from EUTRA to create NAS Container

	TTCN module
	IRAT_Handover_NR5GC_NR

	MCC160 Comment
	Accepted


Before Change

	  function f_NR5GC_508RRC_InterRAT_HO_ToNR(NR_CellId_Type                       p_TargetCellId,

                                           EUTRA_NR_PduSessionInfoList_Type     p_PduSessionInfoList,

                                           RNTI_Value                           p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                           template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit

                                           ) runs on NR5GC_PTC

  {

    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_TargetCellId);

    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig := cs_NR_RachProcedureConfig_CRNTI_HO(v_NR_PhysicalParameters, p_RNTI_Value); //CBRA

    var template (value) RRCReconfiguration v_RRCReconfiguration;

    var template (value) PDUSessionInfoList_Type v_SessionInfoList;

    var integer v_NoOfPduSessions := lengthof(p_PduSessionInfoList);

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var template(value) CellGroupConfig v_CellGroupConfig;

    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)

    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;

    var template (value) MasterKeyUpdate v_MasterKeyUpdate;

    var NextHopChainingCount v_NCC := tsc_NR_38508_NextHopChainingCount;

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var MessageAuthenticationCode v_CalculatedMac;

    var template (value) S1ToN1TransparentContainer v_NASContainer;

    var bitstring v_NASContainerString;

    var integer i;

    if (isvalue(p_RACH_ConfigDedicated.uplink)){ //CFRA

       v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_TargetCellId);

       v_RachProcedureConfig := cs_NR_RachProcedureConfig_CFRA(v_NR_PhysicalParameters);

    }

    v_NASContainer := cs_S1ToN1TransparentContainer (omit,         // FFS should this be set to omit in order to calculate its own value?

                                                     v_SecurityParams.NAS_Ciphering.Algorithm,

                                                     v_SecurityParams.NAS_Integrity.Algorithm,

                                                     int2bit(v_NCC, 3),
                                                     tsc_NasKsi_MappedSecurityContext,

                                                     v_SecurityParams.KSIamf);

    v_NASContainerString := encvalue(valueof(v_NASContainer));

    v_CalculatedMac := fx_NG_NasIntegrityAlgorithm(bit2oct(v_NASContainerString),

                                                   v_SecurityParams.NAS_Integrity.Algorithm,

                                                   v_SecurityParams.NAS_Integrity.K_NAS,

                                                   'FFFFFFFF'O,  // according to 33.501 cl. 8.4.2

                                                   '00001'B,

                                                   tsc_DirectionDL);

                         ……                           

  }


After Change

	function f_NR5GC_508RRC_InterRAT_HO_ToNR(NR_CellId_Type                       p_TargetCellId,

                                           EUTRA_NR_PduSessionInfoList_Type     p_PduSessionInfoList,

                                           RNTI_Value                           p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                           template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,

                                           EUTRA_NR_CoOrd_Security_Type p_EUTRA_Auth_Params 

                                           ) runs on NR5GC_PTC

  {

    var NR_PhysicalParameters_Type v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_TargetCellId);

    var template (value) NR_RachProcedureConfig_Type v_RachProcedureConfig := cs_NR_RachProcedureConfig_CRNTI_HO(v_NR_PhysicalParameters, p_RNTI_Value); //CBRA

    var template (value) RRCReconfiguration v_RRCReconfiguration;

    var template (value) PDUSessionInfoList_Type v_SessionInfoList;

    var integer v_NoOfPduSessions := lengthof(p_PduSessionInfoList);

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var template(value) CellGroupConfig v_CellGroupConfig;

    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)

    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;

    var template (value) MasterKeyUpdate v_MasterKeyUpdate;

    var NextHopChainingCount v_NCC := tsc_NR_38508_NextHopChainingCount;

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var MessageAuthenticationCode v_CalculatedMac;

    var template (value) S1ToN1TransparentContainer v_NASContainer;

    var bitstring v_NASContainerString;

    var integer i;

    if (isvalue(p_RACH_ConfigDedicated.uplink)){ //CFRA

       v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_TargetCellId);

       v_RachProcedureConfig := cs_NR_RachProcedureConfig_CFRA(v_NR_PhysicalParameters);

    }

    v_NASContainer := cs_S1ToN1TransparentContainer (omit,         // FFS should this be set to omit in order to calculate its own value?

                                                     v_SecurityParams.NAS_Ciphering.Algorithm,

                                                     v_SecurityParams.NAS_Integrity.Algorithm,

                                                     //int2bit(v_NCC, 3),

                                                     int2bit(p_EUTRA_Auth_Params.NCC, 3),

                                                     tsc_NasKsi_MappedSecurityContext,

                                                     //v_SecurityParams.KSIamf);

                                                     p_EUTRA_Auth_Params.KSI); 

    v_NASContainerString := encvalue(valueof(v_NASContainer));

    v_CalculatedMac := fx_NG_NasIntegrityAlgorithm(bit2oct(v_NASContainerString),

                                                   v_SecurityParams.NAS_Integrity.Algorithm,

                                                   v_SecurityParams.NAS_Integrity.K_NAS,

                                                   'FFFFFFFF'O,  // according to 33.501 cl. 8.4.2

                                                   '00001'B,

                                                   tsc_DirectionDL);

      ……                                              

      }


2.6 Change 6
	Function name
	fl_TC_8_1_4_2_2_1_TestBody

	Reason for change
	See change 5

	Summary of change
	Add input parameter to pass received security information form EUTRA

	TTCN module
	IRAT_Handover_NR5GC_NR

	MCC160 Comment
	Accepted


Before Change

	  function fl_TC_8_1_4_2_2_1_TestBody() runs on NR5GC_PTC

  {

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var template (omit) GSM_MobilityInfo_Type v_PDUSessionInfo;

    var integer v_DRBId;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var TriggerOrTimeout_Type v_TriggerOrTimeout;

    timer t_ReregisterIMS := 10.0;

    //@siclog "Step 0" siclog@

    //@sic R5-198922 sic@

    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    f_Delay (5.0);

    //@siclog "Steps 1-2" siclog@

    // derive 5G parameters from values received from EUTRA

    f_NR5GC_MappedContextFromS1_InitNAS (nr_Cell1, v_SecurityRx, true);

    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);

    // this also sends octetstring to EUTRA and waits for the Complete message

    f_NR5GC_508RRC_InterRAT_HO_ToNR(nr_Cell1, v_PduSessionInfoList);
f_NR_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

…..

}


After Change

	function fl_TC_8_1_4_2_2_1_TestBody() runs on NR5GC_PTC

  {

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var template (omit) GSM_MobilityInfo_Type v_PDUSessionInfo;

    var integer v_DRBId;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var TriggerOrTimeout_Type v_TriggerOrTimeout;

    timer t_ReregisterIMS := 10.0;

    //@siclog "Step 0" siclog@

    //@sic R5-198922 sic@

    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    f_Delay (5.0);

    //@siclog "Steps 1-2" siclog@

    // derive 5G parameters from values received from EUTRA

    f_NR5GC_MappedContextFromS1_InitNAS (nr_Cell1, v_SecurityRx, true);

    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);

    // this also sends octetstring to EUTRA and waits for the Complete message

    f_NR5GC_508RRC_InterRAT_HO_ToNR(nr_Cell1, v_PduSessionInfoList,-,-,v_SecurityRx);

    f_NR_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

….

}


2.7 Change 7
	Function name
	f_NR5GC_MappedContextFromS1_InitNAS

	Reason for change
	1, See change 1

2, According to 33.501 s8.4.2

NOTE 3:
The target AMF derives a temporary KgNB using the mapped KAMF instead of using the {NH, NCC} pair received from the MME. The uplink NAS COUNT value for the initial KgNB derivation is set to 232-1. The reason for choosing such a value is to avoid any possibility that the value may be used to derive the same KgNB again.

4.
The target AMF requests the target gNB/ng-eNB to establish the bearer(s) by sending the Handover Request message.
The target AMF sends the NAS Container created in step 3 along with, the {NCC=0, NH=derived temporary KgNB}, the New Security Context Indicator (NSCI), and the UE security capabilities in the Handover Request message to the target gNB/ng-eNB.

5.
The target gNB/ng-eNB shall selects the 5G AS security algorithms from the list in the UE security capabilities 
The target gNB/ng-eNB shall compute the KgNB to be used with the UE by performing the key derivation defined in Annex A.11 with the {NCC, NH} pair received in the Handover Request message and the target PCI and its frequency ARFCN-DL. The target gNB/ng-eNB shall associate the NCC value received from AMF with the KgNB.The target gNB /ng-eNB shall then derive the 5G AS security context, by deriving the 5G AS keys (KRRCint, KRRCenc, KUPint, and KUPenc) from the KgNB and the selected 5G AS security algorithm identifiers as described in Annex A.8 for gNB and in Annex A.7 in TS 33.401[10]. 

	Summary of change
	1. derive Kamf using NH from EUTRA

2. AS security key is derived and set in f_NR5GC_508RRC_InterRAT_HO_ToNR_Common_Steps1_14, so here just derive the temporary KgNB, set NCC to zero.


	TTCN module
	NR5GC_IRAT_CommonFunctions

	MCC160 Comment
	Accepted


Before Change

	  function f_NR5GC_MappedContextFromS1_InitNAS(NR_CellId_Type p_CellId,

                                               EUTRA_NR_CoOrd_Security_Type p_EUTRA_Auth_Params,

                                               boolean p_Handover )  runs on NR5GC_PTC

  {

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();

    var template (omit) PhysCellId v_TargetPhysicalCellIdentity := omit;

    var template (omit) ARFCN_ValueNR  v_ARFCN_DL := omit;

    var boolean v_IsSMC_Procedure := true;

    if (p_Handover) {

      v_SecurityParams.KAMF := f_NG_Authentication_A15_HO(v_RRC_SecurityParams.NH,
                                                          p_EUTRA_Auth_Params.Key,

                                                          v_SecurityParams.KDF);

      v_TargetPhysicalCellIdentity := f_NR_CellInfo_GetPhysicalCellId(p_CellId);

      v_ARFCN_DL := f_NR_CellInfo_GetFrequencySSB(p_CellId);

      v_IsSMC_Procedure := false;

    } else {

      v_SecurityParams.KAMF := f_NG_Authentication_A15_IdleMode(p_EUTRA_Auth_Params.NASCountUL,

                                                                p_EUTRA_Auth_Params.Key,

                                                                v_SecurityParams.KDF);

    }

    // Derives KNASenc key

    v_SecurityParams.NAS_Ciphering.K_NAS := f_NG_Authentication_A8 (tsc_NAS_Enc_Alg,v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                           v_SecurityParams.KAMF,

                                                                           v_SecurityParams.KDF);

    // Derives KNASint key

    v_SecurityParams.NAS_Integrity.K_NAS := f_NG_Authentication_A8 (tsc_NAS_Int_Alg,v_SecurityParams.NAS_Integrity.Algorithm,

                                                                           v_SecurityParams.KAMF,

                                                                           v_SecurityParams.KDF);

    v_RRC_SecurityParams := f_NR_Authentication_InitAS (v_RRC_SecurityParams, 'FFFFFFFF'O, v_IsSMC_Procedure, v_TargetPhysicalCellIdentity, v_ARFCN_DL); // @sic R5s200887 sic@
    // Now update the SS

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering);

    f_SS_NG_NAS_CountSet(NASCTRL, '00000000'O, omit);

    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_CellId, false);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_RRC_SecurityParams.AS_Integrity,

                                          v_RRC_SecurityParams.AS_Ciphering);
    f_NR_Security_Set(v_RRC_SecurityParams);

    f_NR5GC_Security_Set (v_SecurityParams);

  }


After Change

	function f_NR5GC_MappedContextFromS1_InitNAS(NR_CellId_Type p_CellId,

                                               EUTRA_NR_CoOrd_Security_Type p_EUTRA_Auth_Params,

                                               boolean p_Handover )  runs on NR5GC_PTC

  {

    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();

    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();

    var template (omit) PhysCellId v_TargetPhysicalCellIdentity := omit;

    var template (omit) ARFCN_ValueNR  v_ARFCN_DL := omit;

    var boolean v_IsSMC_Procedure := true;

    if (p_Handover) {

      v_SecurityParams.KAMF := f_NG_Authentication_A15_HO(p_EUTRA_Auth_Params.NH,

                                                          p_EUTRA_Auth_Params.Key,

                                                          v_SecurityParams.KDF);

      v_TargetPhysicalCellIdentity := f_NR_CellInfo_GetPhysicalCellId(p_CellId);

      v_ARFCN_DL := f_NR_CellInfo_GetFrequencySSB(p_CellId);

      v_IsSMC_Procedure := false;

    } else {

      v_SecurityParams.KAMF := f_NG_Authentication_A15_IdleMode(p_EUTRA_Auth_Params.NASCountUL,

                                                                p_EUTRA_Auth_Params.Key,

                                                                v_SecurityParams.KDF);

    }

    // Derives KNASenc key

    v_SecurityParams.NAS_Ciphering.K_NAS := f_NG_Authentication_A8 (tsc_NAS_Enc_Alg,v_SecurityParams.NAS_Ciphering.Algorithm,

                                                                           v_SecurityParams.KAMF,

                                                                           v_SecurityParams.KDF);

    // Derives KNASint key

    v_SecurityParams.NAS_Integrity.K_NAS := f_NG_Authentication_A8 (tsc_NAS_Int_Alg,v_SecurityParams.NAS_Integrity.Algorithm,

                                                                           v_SecurityParams.KAMF,

                                                                           v_SecurityParams.KDF);

    v_RRC_SecurityParams.KAMF := v_SecurityParams.KAMF;  

    //tmp Kgnb = NH

    v_RRC_SecurityParams.KGNB := f_NR_Authentication_A9(v_RRC_SecurityParams.KDF, v_SecurityParams.KAMF, 'FFFFFFFF'O);

    v_RRC_SecurityParams.NH := v_RRC_SecurityParams.KGNB;

    // NCC is set to default/initial value of 0

    v_RRC_SecurityParams.NCC := 0;
    // Now update the SS

    f_SS_NG_NAS_ConfigureSecurity(NASCTRL, v_SecurityParams.NAS_Integrity, v_SecurityParams.NAS_Ciphering);

    f_SS_NG_NAS_CountSet(NASCTRL, '00000000'O, omit);

    /*REMOVED v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_CellId, false);

    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_CellId,

                                          v_RRC_SecurityParams.AS_Integrity,

                                          v_RRC_SecurityParams.AS_Ciphering);

                                         */
    f_NR_Security_Set(v_RRC_SecurityParams);

    f_NR5GC_Security_Set (v_SecurityParams);

  }


2.8 Change 8
	Function name
	f_NR_GetDefaultDRB_InDRBList

	Reason for change
	If no IMS PDU session is created, the call to f_NR_GetDefaultDRB_ForIMSPDUSession() will actually cause a fail verdict due to the error checking in f_NR_GetDefaultDRB_InDRBList()



	Summary of change
	Remove f_NR_SetVerdictFailOrInconc and return -1, which will be handled in f_NR_GetDefaultDRB_ForIMSPDUSession()

	TTCN module
	NR5GC_CommonFunctions

	MCC160 Comment
	Accepted


Before Change

	function f_NR_GetDefaultDRB_InDRBList (IntegerList_Type p_ListOfDRBs) runs on NR5GC_PTC return integer

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var integer i;

    var integer v_DRBIndex;

    if (lengthof(p_ListOfDRBs) == 1) {

      if (p_ListOfDRBs[0] < 0) { // There's an error

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Error finding DRB id " & int2str(p_ListOfDRBs[0]));
        return -1;

      } else { // only one entry and not an error

        return p_ListOfDRBs[0];

      }

    } else { // more than entry

      v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

      for (i:= 0; i < lengthof(p_ListOfDRBs); i :=i+1) { // find the default DRB

        v_DRBIndex := f_NR_DRBInfo_FindEntry (v_ExistingDRBs, p_ListOfDRBs[i]);

        if (v_ExistingDRBs[v_DRBIndex].DefaultDRB) {

          break;

        }

      }

      return p_ListOfDRBs[i];

    }

  }


After Change

	function f_NR_GetDefaultDRB_InDRBList (IntegerList_Type p_ListOfDRBs) runs on NR5GC_PTC return integer

  {

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var integer i;

    var integer v_DRBIndex;

    if (lengthof(p_ListOfDRBs) == 1) {

      if (p_ListOfDRBs[0] < 0) { // There's an error

        //REMOVED f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Error finding DRB id " & int2str(p_ListOfDRBs[0]));
        return -1;

      } else { // only one entry and not an error

        return p_ListOfDRBs[0];

      }

    } else { // more than entry

      v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

      for (i:= 0; i < lengthof(p_ListOfDRBs); i :=i+1) { // find the default DRB

        v_DRBIndex := f_NR_DRBInfo_FindEntry (v_ExistingDRBs, p_ListOfDRBs[i]);

        if (v_ExistingDRBs[v_DRBIndex].DefaultDRB) {

          break;

        }

      }

      return p_ListOfDRBs[i];

    }

  }


2.9 Change 9
	Function name
	f_BuildRRCReconfigParamsNewDRBs

	Reason for change
	1. The new DRB information is not stored locally, which leads to a run-time error when trying to get existing DRB information later
2.  spCellConfigDedicated is not included in RRCReconfiguration message which will lead to a handover failure

	Summary of change
	1. Save the DRB info created 
2. Include  spCellConfigDedicated 

	TTCN module
	NR5GC_NASSteps

	MCC160 Comment
	1. Accepted in principle, but store it in calling function f_NR5GC_IRAT_S1ToN1_SessionModification
2. Accepted


Before Change

	  function f_BuildRRCReconfigParamsNewDRBs (out template (value) RadioBearerConfig p_RadioBearerConfig,

                                      out template (value) CellGroupConfig p_CellGroupConfig,

                                      out template (value) NR_RadioBearerList_Type p_SSConfig,

                                      out template (value) SDAP_ConfigurationList_Type p_SDAP_Configuration,

                                      NR_CellId_Type p_CellId,

                                      PDUSessionInfoList_Type p_GSM_MobilityInfoList,

                                      SRBs_AlreadyConfigured p_SRBs := SRB1,

                                      template (value) SDAP_Config p_SDAP_Config := cs_SDAP_Config) runs on NR5GC_PTC

  { // Split function so list of DRBs can be built elsewhere @sic R5-199103 sic@

    var template (omit) SRB_ToAddModList v_SRBToAddModList := omit;

    var template (value) DRB_ToAddModList v_DRBToAddModList;

    var integer v_NewDRBId;

    var NR_DRB_IdentityList_Type v_DrbList;

    var boolean v_IsFR1 := f_NR_CellInfo_GetIsFR1(p_CellId);

    var SDAP_Config v_SDAP_Config := valueof(p_SDAP_Config); //@sic R5-190720 sic@

    var integer i := 0;

    var integer j := 0; // for CellGroupConfig

    for (i := 0; i < lengthof(p_GSM_MobilityInfoList); i := i + 1) {

      // Set SDAP PDU session Id to the value received in SessionEstablishmentReq @sic R5-190720 sic@

      v_SDAP_Config.pdu_Session  := oct2int(p_GSM_MobilityInfoList[i].SessionId);  // @sic R5s200564 sic@

      p_SDAP_Configuration[i] := f_ReturnNR_SDAP_Configuration_FromSDAP_Config(v_SDAP_Config);

      // Get the DRB Id to use

      v_NewDRBId := f_NR_GetNextDRBId(p_GSM_MobilityInfoList[i], v_SDAP_Config.defaultDRB);

      v_DrbList[i] := v_NewDRBId;

       //SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here

      p_SSConfig[i] := cs_NR_SS_DRB_Config(v_NewDRBId,

                                           cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsDRB),

                                           cs_NR_RlcBearerConfig_AM(v_IsFR1, -, f_NR_LogicalChannelId_DRB(v_NewDRBId)),

                                           p_SDAP_Configuration[i]);

      v_DRBToAddModList[i] := cs_38508_NRDRB_ToAddMod(v_SDAP_Config, v_NewDRBId, cs_38508_PDCP_Config);

    }

    p_CellGroupConfig := f_NR_GetCellgroupConfig_DRB (p_CellId, v_DrbList, false); // @sic R5s200417 sic@

    j := i;  // j is now the length of the DRB list @sic R5s200417 sic@

    i := 0;

    if (p_SRBs == NONE) { // @sic R5-198923 sic@

      v_SRBToAddModList[i] := cs_NR_SRB_ToAddMod(tsc_SRB1);

      p_CellGroupConfig.rlc_BearerToAddModList[j] := valueof(cs_38508_RLC_BearerConfig_SRB(tsc_SRB1));

      p_CellGroupConfig.mac_CellGroupConfig := f_NR_CellInfo_GetMAC_CellGroupConfig(p_CellId);  // @sic R5-202639 sic@

      p_CellGroupConfig.physicalCellGroupConfig := f_NR_CellInfo_GetPhysicalCellGroupConfig(p_CellId);            // @sic R5-202639 sic@

      p_CellGroupConfig.spCellConfig := f_NR_GetSpCellConfigDef(p_CellId);                      // @sic R5-202639 sic@

      p_CellGroupConfig.spCellConfig.reconfigurationWithSync := cs_NR_ReconfigurationWithSyncDef(f_NR_GetServingCellConfigCommon(p_CellId),

                                                                                               tsc_C_RNTI_Value3,

                                                                                               -, //  p_T304 := ms1000,

                                                                                               omit);

      i := i + 1;

      j := j + 1;

    }

    if (p_SRBs != SRB1_2) { // @sic R5-198923 sic@

      v_SRBToAddModList[i] := cs_NR_SRB_ToAddMod(tsc_SRB2);

      p_CellGroupConfig.rlc_BearerToAddModList[j] := valueof(cs_38508_RLC_BearerConfig_SRB(tsc_SRB2));

    }

    p_RadioBearerConfig := cs_38508_RadioBearerConfigDef(v_SRBToAddModList,

                                                         v_DRBToAddModList,

                                                         -,

                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));

  }


After Change

	  function f_BuildRRCReconfigParamsNewDRBs (out template (value) RadioBearerConfig p_RadioBearerConfig,

                                      out template (value) CellGroupConfig p_CellGroupConfig,

                                      out template (value) NR_RadioBearerList_Type p_SSConfig,

                                      out template (value) SDAP_ConfigurationList_Type p_SDAP_Configuration,

                                      NR_CellId_Type p_CellId,

                                      PDUSessionInfoList_Type p_GSM_MobilityInfoList,

                                      SRBs_AlreadyConfigured p_SRBs := SRB1,

                                      template (value) SDAP_Config p_SDAP_Config := cs_SDAP_Config) runs on NR5GC_PTC

  { // Split function so list of DRBs can be built elsewhere @sic R5-199103 sic@

    var template (omit) SRB_ToAddModList v_SRBToAddModList := omit;

    var template (value) DRB_ToAddModList v_DRBToAddModList;

    var integer v_NewDRBId;

    var NR_DRB_IdentityList_Type v_DrbList;

    var boolean v_IsFR1 := f_NR_CellInfo_GetIsFR1(p_CellId);

    var SDAP_Config v_SDAP_Config := valueof(p_SDAP_Config); //@sic R5-190720 sic@

    var integer i := 0;

    var integer j := 0; // for CellGroupConfig

    var NR_DRBInfoList_Type v_ExistingDRBs;

    var NR_DRBInfoList_Type v_NewDRBInfoList;

    var integer v_NoOfDRBs := f_NR5GC_MobileInfo_GetNoOfDRBs();
    for (i := 0; i < lengthof(p_GSM_MobilityInfoList); i := i + 1) {

      // Set SDAP PDU session Id to the value received in SessionEstablishmentReq @sic R5-190720 sic@

      v_SDAP_Config.pdu_Session  := oct2int(p_GSM_MobilityInfoList[i].SessionId);  // @sic R5s200564 sic@

      p_SDAP_Configuration[i] := f_ReturnNR_SDAP_Configuration_FromSDAP_Config(v_SDAP_Config);

      // Get the DRB Id to use

      v_NewDRBId := f_NR_GetNextDRBId(p_GSM_MobilityInfoList[i], v_SDAP_Config.defaultDRB);

      v_DrbList[i] := v_NewDRBId;

       //SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here

      p_SSConfig[i] := cs_NR_SS_DRB_Config(v_NewDRBId,

                                           cs_NR_PDCP_Configuration_RBTerminating(cs_NR_PDCP_RbConfig_ParamsDRB),

                                           cs_NR_RlcBearerConfig_AM(v_IsFR1, -, f_NR_LogicalChannelId_DRB(v_NewDRBId)),

                                           p_SDAP_Configuration[i]);

      v_DRBToAddModList[i] := cs_38508_NRDRB_ToAddMod(v_SDAP_Config, v_NewDRBId, cs_38508_PDCP_Config);

    }

    p_CellGroupConfig := f_NR_GetCellgroupConfig_DRB (p_CellId, v_DrbList, false); // @sic R5s200417 sic@

    j := i;  // j is now the length of the DRB list @sic R5s200417 sic@

    i := 0;

    if (p_SRBs == NONE) { // @sic R5-198923 sic@

      v_SRBToAddModList[i] := cs_NR_SRB_ToAddMod(tsc_SRB1);

      p_CellGroupConfig.rlc_BearerToAddModList[j] := valueof(cs_38508_RLC_BearerConfig_SRB(tsc_SRB1));

      p_CellGroupConfig.mac_CellGroupConfig := f_NR_CellInfo_GetMAC_CellGroupConfig(p_CellId);  // @sic R5-202639 sic@

      p_CellGroupConfig.physicalCellGroupConfig := f_NR_CellInfo_GetPhysicalCellGroupConfig(p_CellId);            // @sic R5-202639 sic@

      p_CellGroupConfig.spCellConfig := f_NR_GetSpCellConfigDef(p_CellId, f_NR_GetServinCellConfig(p_CellId));// @sic R5-202639 sic@

      p_CellGroupConfig.spCellConfig.reconfigurationWithSync := cs_NR_ReconfigurationWithSyncDef(f_NR_GetServingCellConfigCommon(p_CellId),

                                                                                               tsc_C_RNTI_Value3,

                                                                                               -, //  p_T304 := ms1000,

                                                                                               omit);

      i := i + 1;

      j := j + 1;

    }

    if (p_SRBs != SRB1_2) { // @sic R5-198923 sic@

      v_SRBToAddModList[i] := cs_NR_SRB_ToAddMod(tsc_SRB2);

      p_CellGroupConfig.rlc_BearerToAddModList[j] := valueof(cs_38508_RLC_BearerConfig_SRB(tsc_SRB2));

    }

    p_RadioBearerConfig := cs_38508_RadioBearerConfigDef(v_SRBToAddModList,

                                                         v_DRBToAddModList,

                                                         -,

                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));

    v_NewDRBInfoList := fl_BuildDRBNewDRBList (valueof(v_DRBToAddModList), valueof(p_SSConfig), valueof(p_CellGroupConfig));

    if (v_NoOfDRBs > 0) {

      v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();

      v_ExistingDRBs := f_NR_DRBInfo_AddListOfEntries (v_ExistingDRBs, v_NewDRBInfoList);

    } else {

      v_ExistingDRBs := v_NewDRBInfoList;

    }

    // Store updated DRB information

    f_NR5GC_MobileInfo_SetDRBInfoList(v_ExistingDRBs);
  }


2.10 Change 10
	Function name
	fl_TC_8_1_4_2_2_1_TestBody

	Reason for change
	1.PiggyBackMsgList is included in the UL NAS message at step3
2. DRB routing table should be updated to indicate raw mode handling so that IP packets received at step 7 are sent to NR PTC DRB port.

3. DRB ID is currently hardcoded

	Summary of change
	1. change expected UL message
2. change IP routing to raw mode for datapath checking

3. Use variable for DRB ID

	TTCN module
	IRAT_Handover_NR5GC_NR

	MCC160 Comment
	1 Accepted
2. Not relevant according to agreed R5-204453


Before Change

	  function fl_TC_8_1_4_2_2_1_TestBody() runs on NR5GC_PTC

  {

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var template (omit) GSM_MobilityInfo_Type v_PDUSessionInfo;

    var integer v_DRBId;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var TriggerOrTimeout_Type v_TriggerOrTimeout;

    timer t_ReregisterIMS := 10.0;

    //@siclog "Step 0" siclog@

    //@sic R5-198922 sic@

    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    f_Delay (5.0);

    //@siclog "Steps 1-2" siclog@

    // derive 5G parameters from values received from EUTRA

    f_NR5GC_MappedContextFromS1_InitNAS (nr_Cell1, v_SecurityRx, true);

    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);

    // this also sends octetstring to EUTRA and waits for the Complete message

    f_NR5GC_508RRC_InterRAT_HO_ToNR(nr_Cell1, v_PduSessionInfoList,-,-,v_SecurityRx);

    f_NR_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

    //@siclog "Steps 3-5" siclog@

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected))) -> value v_ReceivedAsp;

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease, -, tsc_NR_RbId_SRB2);

    // Procedure for UE-requested PDU session modification according to 38.508 Table 4.9.14.2.2-1

    f_NR5GC_IRAT_S1ToN1_SessionModification(nr_Cell1);

    //  Deal with potential IMS Re-Registration

    if (f_NR_GetDefaultDRB_ForIMSPDUSession() > -1) { // There is an IMS session

      v_TriggerOrTimeout := f_IMS_IPCAN_WaitForTriggerOrTimeout(IMS[tsc_Index_PDN1], t_ReregisterIMS);

      select (v_TriggerOrTimeout) {

        case (TRIGGER) {                                          // => IMS Re-Registration has been started ...

          f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);        //    ... wait for IMS Re-Registration being finished

        }

        case (TIMEOUT) {

          f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1], cms_IPCAN_IMS_ReregistrationStop);

        }

      }

    }

    // @siclog "Steps 7" siclog@

    // Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1 indicate that the UE is capable of exchanging IP data?

    // Prose needs to be updated to remove hard-coding of DRB1 FFS

    v_DRBId := f_NR_GetDefaultDRB_ForFirstNonIMSPDUSession();

    v_PDUSessionInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);

     if (isvalue(v_PDUSessionInfo)) { // @sic R5s200517 sic@

      f_NR5GC_CheckDataPath(nr_Cell1, f_NR5GC_GetPdnIndex(valueof(v_PDUSessionInfo)), dedicatedBearer, 1, "Step 7");

     } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No Internet PDU Session");

    }

 }


After Change

	  function fl_TC_8_1_4_2_2_1_TestBody() runs on NR5GC_PTC

  {

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (EUTRA);

    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var template (omit) GSM_MobilityInfo_Type v_PDUSessionInfo;

    var integer v_DRBId;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var TriggerOrTimeout_Type v_TriggerOrTimeout;

    timer t_ReregisterIMS := 10.0;

    //@siclog "Step 0" siclog@

    //@sic R5-198922 sic@

    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    f_Delay (5.0);

    //@siclog "Steps 1-2" siclog@

    // derive 5G parameters from values received from EUTRA

    f_NR5GC_MappedContextFromS1_InitNAS (nr_Cell1, v_SecurityRx, true);

    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(EUTRA);

    // this also sends octetstring to EUTRA and waits for the Complete message

    f_NR5GC_508RRC_InterRAT_HO_ToNR(nr_Cell1, v_PduSessionInfoList,-,-,v_SecurityRx);

    f_NR_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

    //@siclog "Steps 3-5" siclog@

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected))) -> value v_ReceivedAsp;

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease, -, tsc_NR_RbId_SRB2);

    // Procedure for UE-requested PDU session modification according to 38.508 Table 4.9.14.2.2-1

    f_NR5GC_IRAT_S1ToN1_SessionModification(nr_Cell1);

    //  Deal with potential IMS Re-Registration

    if (f_NR_GetDefaultDRB_ForIMSPDUSession() > -1) { // There is an IMS session

      v_TriggerOrTimeout := f_IMS_IPCAN_WaitForTriggerOrTimeout(IMS[tsc_Index_PDN1], t_ReregisterIMS);

      select (v_TriggerOrTimeout) {

        case (TRIGGER) {                                          // => IMS Re-Registration has been started ...

          f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);        //    ... wait for IMS Re-Registration being finished

        }

        case (TIMEOUT) {

          f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1], cms_IPCAN_IMS_ReregistrationStop);

        }

      }

    }

    f_IP_Routing_ChangeToRawmode(IP);
    // @siclog "Steps 7" siclog@

    // Check: Does the test result of generic test procedure in TS 38.508-1 subclause 4.9.1 indicate that the UE is capable of exchanging IP data?

    // Prose needs to be updated to remove hard-coding of DRB1 FFS

    v_DRBId := f_NR_GetDefaultDRB_ForFirstNonIMSPDUSession();

    v_PDUSessionInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Internet_DNN);

     if (isvalue(v_PDUSessionInfo)) { // @sic R5s200517 sic@

      f_NR5GC_CheckDataPath(nr_Cell1, f_NR5GC_GetPdnIndex(valueof(v_PDUSessionInfo)), dedicatedBearer, v_DRBId, "Step 7");

     } else {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No Internet PDU Session");

    }

 }


